Práctica 4. Integración Numérica
Análise Matemática. Grao en Enxeñería Informática.
E.S.E.I Ourense. Universidade de Vigo.

1 *Formulas de tipo interpolatorio*

Figure 1:

Fórmulas de tipo interpolatorio

1) Tomamos n+1 puntos distintos, x_i, $i = 0, 1, \ldots, n$, del intervalo [a,b]
2) Calculamos el polinomio de interpolación de la función f en los puntos x_i
3) Aproximamos la integral de la función por la integral del polinomio de interpolación

$$\int_a^b f(x)\,dx \simeq \int_a^b p(x)\,dx.$$

Veamos como se obtienen las formulas del trapecio (interpolando en los puntos (a,f(a)) y (b,f(b)))
y de Simpson (interpolando en los puntos (a,f(a)), ((a+b)/2,f((a+b)/2)) y (b,f(b))).

1.1 Formula del Trapecio

```
(%i1) kill(all);
      load(interpol);
      datos:[[a,f(a)],[b,f(b)]];
      lagrange(datos);
```

(%o0) *done*

(%o1) */Applications/Maxima.app/Contents/Resources/maxima/share/maxima/5.25.1/share/numeric/interpol.mac*

(%o2) $[[a, f(a)], [b, f(b)]]$

(%o3) $\dfrac{f(a)\,(x-b)}{a-b} + \dfrac{f(b)\,(x-a)}{b-a}$

```
(%i4) integrate(lagrange(datos), x, a, b);
```

(%o4) $\dfrac{(b^2-2\,a\,b)\,f(b)+f(a)\,b^2}{2\,b-2\,a} + \dfrac{a^2\,f(b)-2\,a\,f(a)\,b+a^2\,f(a)}{2\,b-2\,a}$

```
(%i5) factor(%);
```

(%o5) $\dfrac{(b-a)\,(f(b)+f(a))}{2}$

1.2 Formula de Simpson

```
(%i6) kill(all);
      load(interpol);
      datos:[[a,f(a)],[(a+b)/2,f((a+b)/2)],[b,f(b)]];
      lagrange(datos);
```

(%o0) *done*

(%o1) */Applications/Maxima.app/Contents/Resources/maxima/share/maxima/5.25.1/share/numeric/interpol.mac*

(%o2) $\left[[a, f(a)], \left[\dfrac{b+a}{2}, f\left(\dfrac{b+a}{2}\right)\right], [b, f(b)]\right]$

(%o3) $\dfrac{f(a)\,(x-b)\left(x-\dfrac{b+a}{2}\right)}{(a-b)\left(a-\dfrac{b+a}{2}\right)} + \dfrac{f(b)\,(x-a)\left(x-\dfrac{b+a}{2}\right)}{(b-a)\left(b-\dfrac{b+a}{2}\right)} + \dfrac{f\left(\dfrac{b+a}{2}\right)(x-a)\,(x-b)}{\left(\dfrac{b+a}{2}-a\right)\left(\dfrac{b+a}{2}-b\right)}$

(%i4) integrate(lagrange(datos), x, a, b);

(%o4) $$\frac{\left(4\,b^3-12\,a\,b^2\right)\,f\left(\frac{b+a}{2}\right)+\left(b^3-3\,a\,b^2+6\,a^2\,b\right)\,f(b)+f(a)\,b^3+3\,a\,f(a)\,b^2}{6\,b^2-12\,a\,b+6\,a^2}+\frac{\left(12\,a^2\,b-4\,a^3\right)\,f\left(\frac{b+a}{2}\right)+\left(-3\,a^2\,b-a^3\right)\,f(b)-6\,a\,f(a)\,b^2+3\,a^2\,f(a)\,b-a^3\,f(a)}{6\,b^2-12\,a\,b+6\,a^2}$$

(%i5) factor(%);

(%o5) $$\frac{(b-a)\left(4\,f\left(\frac{b+a}{2}\right)+f(b)+f(a)\right)}{6}$$

2 Formulas de integracion simple.

Figure 2:

Fórmula del rectángulo izquierda:

$$\int_a^b f(x)\,dx \simeq f(a)\,(b-a)$$

Fórmula del rectángulo derecha:

$$\int_a^b f(x)\,dx \simeq f(b)\,(b-a)$$

Fórmula del punto medio:

$$\int_a^b f(x)\,dx \simeq f\left(\frac{a+b}{2}\right)(b-a)$$

Fórmula del trapecio:

$$\int_a^b f(x)\,dx \simeq \frac{b-a}{2}\,(f(a)+f(b))$$

Fórmula de Simpson:

$$\int_a^b f(x)\,dx \simeq \frac{b-a}{6}\left[f(a)+4\,f\left(\frac{a+b}{2}\right)+f(b)\right]$$

2.1 Calcular un valor aproximado de la integral de e^x en [0,1/2] aplicando la formula del punto medio. Comparar con el valor exacto.

Definimos la funcion:

```
(%i6) kill(all);
      f(x):=%e^x;
```

(%o0) *done*

(%o1) $f(x):=\%e^x$

Definimos la formula del punto medio:

```
(%i2) PuntoMedio(f,a,b):=block(
      (b-a)*f((a+b)/2)
      );
```

(%o2) $\mathrm{PuntoMedio}(f,a,b):=\mathrm{block}\left((b-a)\,\mathrm{f}\left(\frac{a+b}{2}\right)\right)$

Valor aproximado:

```
(%i3) PuntoMedio(f,0,1/2);
```

(%o3) $\frac{\%e^{1/4}}{2}$

```
(%i4) float(%);
```

(%o4) .6420127083438707

Valor exacto:

```
(%i5) integrate(f(x), x, 0, 1/2);
```

(%o5) $\sqrt{\%e}-1$

```
(%i6) float(%);
```

(%o6) .6487212707001282

Error:

```
(%i7) abs(PuntoMedio(f,0,1/2)-integrate(f(x), x, 0, 1/2));
```

(%o7) $\sqrt{\%e}-\frac{\%e^{1/4}}{2}-1$

```
(%i8) float(%);
```

(%o8) .006708562356257497

2.2 Calcular un valor aproximado de la integral de e^(-x^2) entre 0 y 1, aplicando las formulas del trapecio y de Simpson.

Definimos la funcion:

```
(%i9) kill(all);
      f(x):=%e^(-x^2);
```

(%o0) *done*

(%o1) $\mathrm{f}(x):=\%e^{-x^2}$

Definimos la formula del trapecio y de Simpson:

```
(%i2) Trapecio(f,a,b):=block(
      (b-a)*(f(a)+f(b))/2
      );
```

(%o2) $\mathrm{Trapecio}(f,a,b):=\mathrm{block}\left(\frac{(b-a)\,(\mathrm{f}(a)+\mathrm{f}(b))}{2}\right)$

```
(%i3) Simpson(f,a,b):=block(
      (b-a)*(f(a)+4*f((a+b)/2)+f(b))/6
      );
```

(%o3) $\mathrm{Simpson}(f,a,b):=\mathrm{block}\left(\frac{(b-a)\left(\mathrm{f}(a)+4\,\mathrm{f}\left(\frac{a+b}{2}\right)+\mathrm{f}(b)\right)}{6}\right)$

Valor aproximado:

(%i4) Trapecio(f,0,1);

(%o4) $\frac{\%e^{-1}+1}{2}$

(%i5) float(%);

(%o5) .6839397205857212

(%i6) Simpson(f,0,1);

(%o6) $\frac{4\ \%e^{-\frac{1}{4}}+\%e^{-1}+1}{6}$

(%i7) float(%);

(%o7) .7471804289095103

Valor exacto:

(%i8) integrate(f(x), x, 0, 1);

(%o8) $\frac{\sqrt{\pi}\ \mathrm{erf}(1)}{2}$

(%i9) float(%);

(%o9) .7468241328124269

Error de la formula del trapecio:

(%i10) abs(Trapecio(f,0,1)-integrate(f(x), x, 0, 1));

(%o10) $\frac{\sqrt{\pi}\ \mathrm{erf}(1)}{2}-\frac{\%e^{-1}+1}{2}$

(%i11) float(%);

(%o11) .06288441222670571

Error de la formula de Simpson:

(%i12) abs(Simpson(f,0,1)-integrate(f(x), x, 0, 1));

(%o12) $\frac{4\ \%e^{-\frac{1}{4}}+\%e^{-1}+1}{6}-\frac{\sqrt{\pi}\ \mathrm{erf}(1)}{2}$

(%i13) float(%);

(%o13) $3.562960970834306\ 10^{-4}$

3 Formulas de integracion compuesta

Figure 3:

Consiste en dividir el intervalo inicial en subintervalos y aplicar un método de integración numérica simple en cada uno de ellos. Si llamamos tomamos $h = \frac{b-a}{n}$, entonces los puntos $x_i = a + i\,h,\ i = 0,\ 1,\ \ldots,\ n$ forman una partición del intervalo [a, b] y basta aplicar en cada subintervalo $[x_i,\ x_{i+1}]$, $i = 0,\ 1,\ \ldots,\ n-1$, el método simple que queramos

Fórmula del rectángulo izquierda compuesta:

$$\int_a^b f(x)\,dx \simeq h \sum_{i=0}^{n-1} f_i$$

Fórmula del rectángulo derecha compuesta:

$$\int_a^b f(x)\,dx \simeq h \sum_{i=0}^{n-1} f_{i+1}$$

Fórmula del punto medio compuesta:

$$\int_a^b f(x)\,dx \simeq h \sum_{i=0}^{n-1} f_{i+\frac{1}{2}}$$

Fórmula del trapecio compuesta:

$$\int_a^b f(x)\,dx \simeq \frac{h}{2}\left(f_0 + 2\sum_{i=1}^{n-1} f_i + f_n\right)$$

Fórmula de Simpson compuesta:

$$\int_a^b f(x)\,dx \simeq \frac{h}{6}\left(f_0 + f_n + 2\sum_{i=1}^{n-1} f_i + 4\sum_{i=0}^{n-1} f_{i+\frac{1}{2}}\right)$$

3.1 Calcular un valor aproximado de la integral de la funcion f(x)=(x^3-x+2)*e^x en el intervalo [0,2] aplicando formulas del rectangulo derecha, del trapecio y de Simpson compuestas con n=20.

Definimos la funcion:

```
(%i14) kill(all);
       f(x):=(x^3-x+2)*%e^x;
(%o0) done
```

(%o1) $f(x) := (x^3 - x + 2)\,\%e^x$

Definimos los puntos del intervalo que lo dividen en 20 subintervalos iguales:

Valor exacto:

```
(%i2) integrate(f(x), x, 0, 2);
```

(%o2) $3\,\%e^2 + 3$

```
(%i3) valorexacto:float(%);
(%o3) 25.16716829679195
```

Aproximacion usando la formula del rectangulo derecha compuesta:

```
(%i4) RectanguloDerechaCompuesta(f,a,b,n):=block([sum:0],
      h:(b-a)/n,
      for k:1 thru n do sum:sum+f(a+k*h),
      float(h*sum)
      );
```

(%o4) RectanguloDerechaCompuesta(f,a,b,n):=

block$\left([sum:0],\, h:\frac{b-a}{n},\, \text{for } k \text{ thru } n \text{ do } sum: sum+f(a+k\,h),\, \text{float}(h\, sum)\right)$

```
(%i5) aprox1:RectanguloDerechaCompuesta(f,0,2,20);
```

(%o5) 28.13886636377141

```
(%i6) abs(valorexacto-aprox1);
```

(%o6) 2.971698066979464

Aproximacion usando la formula del trapecio compuesta:

```
(%i7) TrapecioCompuesta(f,a,b,n):=block([sum:0],
      h:(b-a)/n,
      sum:f(a)+f(b),
      for k:1 thru n-1 do sum:sum+2*f(a+k*h),
      float((h/2)*sum)
      );
```

(%o7) TrapecioCompuesta(f,a,b,n):=

block$\left([sum:0],\, h:\frac{b-a}{n},\, sum: f(a)+f(b),\, \text{for } k \text{ thru } n-1 \text{ do } sum: sum+2\, f(a+k\,h),\, \text{float}\left(\frac{h}{2}\, sum\right)\right)$

```
(%i8) aprox2:TrapecioCompuesta(f,0,2,20);
```

(%o8) 25.28324392419916

```
(%i9) abs(valorexacto-aprox2);
```

(%o9) .1160756274072057

Aproximacion usando la formula de Simpson compuesta:

```
(%i10) SimpsonCompuesta(f,a,b,n):=block([sum:0],
       h:(b-a)/n,
       sum:f(a)+f(b),
       for k:1 thru n-1 do sum:sum+2*f(a+k*h),
       for k:0 thru n-1 do sum:sum+4*f(a+(k+1/2)*h),
       float((h/6)*sum)
       );
```

(%o10) SimpsonCompuesta(f,a,b,n):=block ($[sum:0],\, h:\frac{b-a}{n},\, sum: f(a)+f(b),\, \text{for } k \text{ thru } n-1 \text{ do } sum: sum+2$

$f(a+k\,h),\, \text{for } k \text{ from } 0 \text{ thru } n-1 \text{ do } sum: sum+4\, f\left(a+\left(k+\frac{1}{2}\right) h\right),\, \text{float}\left(\frac{h}{6}\, sum\right)$)

```
(%i11) aprox3:SimpsonCompuesta(f,0,2,20);
```

(%o11) 25.16718940007435

```
(%i12) abs(valorexacto-aprox3);
```

(%o12) $2.110328239623982\ 10^{-5}$

3.2 Calcular un valor aproximado de la integral de la funcion f(x)=ln(x) en el intervalo [1,2] aplicando las formulas de integracion del rectangulo izquierda, del trapecio y de Simpson compuestas con n=200.

Definimos la funcion:

```
(%i13) kill(all);
       f(x):=log(x);
```
(%o0) *done*

(%o1) $f(x):=\log(x)$

Valor exacto:

```
(%i2) integrate(f(x), x, 1, 2);
```
(%o2) $2\log(2)-1$

```
(%i3) valorexacto:float(%);
```
(%o3) .3862943611198906

Aproximacion usando la formula del rectangulo izquierda compuesta:

```
(%i4) RectanguloIzquierdaCompuesta(f,a,b,n):=block([sum:0],
      h:(b-a)/n,
      for k:0 thru n-1 do sum:sum+f(a+k*h),
      float(h*sum)
      );
```
(%o4) $\mathrm{RectanguloIzquierdaCompuesta}(f,a,b,n):=$

$$\mathrm{block}\left([sum:0],\,h:\frac{b-a}{n},\,\mathrm{for}\ \ k\ \ \mathrm{from}\ \ 0\ \ \mathrm{thru}\ \ n-1\ \ \mathrm{do}\ \ sum:sum+f(a+k\,h),\,\mathrm{float}(h\,sum)\right)$$

```
(%i5) aprox1:RectanguloIzquierdaCompuesta(f,1,2,200);
```
(%o5) .3845604515033433

```
(%i6) abs(valorexacto-aprox1);
```
(%o6) .001733909616547291

Aproximacion usando la formula del trapecio compuesta:

```
(%i7) TrapecioCompuesta(f,a,b,n):=block([sum:0],
      h:(b-a)/n,
      sum:f(a)+f(b),
      for k:1 thru n-1 do sum:sum+2*f(a+k*h),
      float((h/2)*sum)
      );
```
(%o7) $\mathrm{TrapecioCompuesta}(f,a,b,n):=$

$$\mathrm{block}\left([sum:0],\,h:\frac{b-a}{n},\,sum:f(a)+f(b),\,\mathrm{for}\ \ k\ \ \mathrm{thru}\ \ n-1\ \ \mathrm{do}\ \ sum:sum+2\,f(a+k\,h),\,\mathrm{float}\left(\frac{h}{2}\,sum\right)\right)$$

```
(%i8) aprox2:TrapecioCompuesta(f,1,2,200);
```
(%o8) .3862933194547432

```
(%i9) abs(valorexacto-aprox2);
```
(%o9) $1.0416651474165484\ 10^{-6}$

Aproximacion usando la formula de Simpson compuesta:

```
(%i10) SimpsonCompuesta(f,a,b,n):=block([sum:0],
       h:(b-a)/n,
       sum:f(a)+f(b),
       for k:1 thru n-1 do sum:sum+2*f(a+k*h),
       for k:0 thru n-1 do sum:sum+4*f(a+(k+1/2)*h),
       float((h/6)*sum)
       );
```
(%o10) $\mathrm{SimpsonCompuesta}(f,a,b,n):=\mathrm{block}\,([sum:0],\,h:\frac{b-a}{n},\,sum:f(a)+f(b),\,\mathrm{for}\ \ k\ \ \mathrm{thru}\ \ n-1\ \ \mathrm{do}\ \ sum:sum+2$

$$f(a+k\,h),\,\mathrm{for}\ \ k\ \ \mathrm{from}\ \ 0\ \ \mathrm{thru}\ \ n-1\ \ \mathrm{do}\ \ sum:sum+4\,f\left(a+\left(k+\frac{1}{2}\right)h\right),\,\mathrm{float}\left(\frac{h}{6}\,sum\right))$$

```
(%i11) aprox3:SimpsonCompuesta(f,1,2,200);
(%o11) 0.386294361119511
```

```
(%i12) abs(valorexacto-aprox3);
```

(%o12) $3.79585252119341\ 10^{-13}$